


Hochschulrechenzentrum
Justus-Liebig-Universität Gießen



Word für Microsoft 365

Tabellenberechnungen



Inhaltsverzeichnis

Einleitung.....	2
Berechnungen direkt in einer Tabelle durchführen	2
Zahlenformate festlegen	5
Ergebnisse einer Berechnung außerhalb der Tabelle einsetzen	7
Logische Funktionen.....	8
Alle Funktionen im Überblick	10
Komplexe Funktionen in Tabellen verwenden.....	11

Abbildungsverzeichnis

Abb. 1: <i>Dialogfeld Formel</i>	4
Abb. 2: <i>Zahlenformate im Dialogfeld Formel</i>	5
Abb. 3: <i>Dialogfeld Textmarke</i>	7
Abb. 4: <i>Dialogfeld Querverweis</i>	8
Abb. 5: <i>Excel-Tabelle im Word-Dokument, in der Bearbeitung und nach der Bearbeitung</i>	12

Einleitung

Berechnungen können nicht nur in dafür vorgesehenen Programmen, wie z.B. in dem Tabellenkalkulationsprogramm Excel, durchgeführt werden, sondern auch in Word. Genauer gesagt: in Tabellen in Word¹. Allerdings ist die Auswahl an Funktionen in Word deutlich eingeschränkt. Wenn Sie aber mit den Grundrechenarten und ein paar wenigen Funktionen zufrieden sind, können Sie auch ohne externe Programme Berechnungen in Word durchführen. Das Skript bezieht sich auf [Word für Microsoft 365](#), ist aber auch für die älteren Versionen gültig (auf alle Fälle ab Version 2016).

Berechnungen direkt in einer Tabelle durchführen

Tabellen in Word dienen nicht nur zur Auflistung von mehrspaltigen Textteilen (z.B. Adresslisten), sondern auch zur Darstellung von größeren Zahlenangaben (z.B. Messwerte oder statistische Daten). Gerade in Tabellen mit Zahlenwerten müssen häufig Berechnungen (z.B. die Summe oder der Mittelwert einer Zahlenreihe) durchgeführt werden. In Word gibt es eine (kleine) Reihe von Berechnungsmöglichkeiten. Eine genaue Beschreibung der einzelnen Schritte finden Sie auf Seite 4. Zunächst muss für die Durchführung einer Berechnung der Aufbau einer Word-Tabelle etwas näher erläutert werden: Eine Tabelle ist i. Allg. in Zeilen und Spalten aufgebaut. Analog zu den gängigsten Tabellenkalkulationsprogrammen (z.B. Microsoft Excel) werden die Zeilen mit Nummern und die Spalten mit Buchstaben des Alphabets gekennzeichnet (siehe Beispiele in Tabelle 1). Diese Zahlen (hier gelb dargestellt) und Buchstaben (hier hellorange dargestellt) sind aber in Word „unsichtbar“ (und können auch nicht sichtbar gemacht werden; im Gegensatz zu Excel beispielsweise).

	A	B	C	D	E	F	G	H
1	A1	B1	C1	D1	E1	F1	G1	H1
2	A2	B2	C2	D2	E2	F2	G2	H2
3	A3	B3	C3	D3	E3	F3	G3	H3

Tabelle 1: Die (unsichtbaren) Namen der einzelnen Tabellenzellen

Durch Kombination von Buchstaben² und Nummern kann jede einzelne Zelle einer Tabelle exakt angegeben werden (siehe Tabelle 1). Mittels dieser Zellennamen können nun Berechnungen innerhalb einer Tabelle durchgeführt werden. Dabei repräsentiert jeder Zellname den Inhalt der entsprechenden Zelle. Beispielsweise könnte eine Berechnung „umgangssprachlich“ so formuliert sein:

Berechne die Summe der Werte aus den Tabellenzellen A1, A4, B3, C5 und D2

In Word hat diese Berechnung folgendes Aussehen:

$$=A1+A4+B3+C5+D2$$

¹ Das ist nicht ganz korrekt. Sie können in Word auch Berechnungen ohne Tabellen durchführen. Dazu benötigen Sie allerdings Feldfunktionen (siehe Skript [Word für Microsoft 365 - Feldfunktionen](#)).

² Es gibt keinen Unterschied zwischen Groß- und Kleinbuchstaben

Für den Aufbau einer Formel in Word gibt es folgende Regeln (diese Regeln sind z.T. identisch mit denen aus dem Tabellenkalkulationsprogramm Microsoft Excel):

- Eine Formel beginnt ohne Ausnahme mit einem Gleichheitszeichen (=).
- Die durchzuführende Berechnung wird unmittelbar hinter dem Gleichheitszeichen eingegeben. Dabei können sowohl konstante Werte als auch Zellnamen verwendet werden.
- Für einfache Berechnungen können die Zeichen für Addition (+), Subtraktion (-), Multiplikation (*), Division (/) und Potenzieren (^) verwendet werden. Es gilt die übliche Punkt-vor-Strichrechnung, also gegebenenfalls runde Klammern verwenden (z.B. bedeutet die Formel **=(A1+A2)*B1**, dass zunächst die Inhalte der Tabellenzellen **A1** und **A2** addiert werden und dann die Summe mit dem Inhalt der Tabellenzelle **B1** multipliziert wird).
- Für weitere Berechnungen (z.B. Mittelwert) stehen Funktionen zur Verfügung. Eine Funktion besteht aus einem Namen (z.B. **SUM**), gefolgt von einem runden Klammerpaar. Zwischen den Klammern können konstante Werte, Zellennamen oder andere Funktionsnamen stehen, die als Argumente einer Funktion bezeichnet werden und festlegen, was eigentlich berechnet werden soll. Das Klammerpaar muss bei einer Funktion immer angegeben werden, selbst dann, wenn keine Angaben zwischen den Klammern benötigt werden. **Ausnahme:** die Funktionen **FALSCH** und **WAHR**. Ohne Bedeutung ist die Wahl der Groß-/Kleinschreibweise bei Funktionsnamen (z.B. ist **SUM** identisch mit **Sum** oder **sum**).
- Innerhalb des Klammerpaars werden mehrere Angaben jeweils durch Semikola voneinander getrennt (z.B. Tabellenzellen sind verstreut über die Tabelle). Zusammenhängende Zellbereiche (z.B. **A1** bis **A7**) können mittels eines Doppelpunkts angegeben werden (z.B. **A1:A7**).

In der nachfolgenden Übersicht sind einige gültige und ungültige Formeln als Beispiele angegeben (in dieser Übersicht wird vorausgesetzt, dass bei Angaben von Zellennamen diese auch existieren, also die Tabelle eine Mindestanzahl an Zeilen und Spalten besitzt; ansonsten würden die Beispiele keinen Sinn machen):

Gültige Formeln	Ungültige Formeln	Erklärung
=a3+b7	=3a4*d2+g5	3a4 ist kein Zellename korrekt: =a4*d2+g5
=(12+d4)*(34-d2)	(a2-b4)*67	Gleichheitszeichen fehlt korrekt: =(a2-b4)*67
=sum(a3;c4;b2;d5)	=product(12;a3;)	Letztes Semikolon zu viel korrekt: =product(12;a3)
=4,55*product(a2:b4)	=sum(12:c4)	12:c4 ist kein Zellbereich korrekt: =sum(12;c4)
=a4*sum(12;a4:a8)	=sum(a4+c3)	Doppelte Rechnung ungültig korrekt: =sum(a4;c3) oder =a4+c3
=sum(min(a2;a4);b3)	=max(a1:b8)	Schließende Klammer fehlt korrekt: =max(a1:b8)

Nach dieser kurzen Einführung nun die konkrete Durchführung anhand eines kleinen Beispiels. Hierzu sei folgende Tabelle vorgegeben:

	A	B	C	D
1	12,54653	4,65747	7,34936	-12,65042
2	321,60653	12,23254	0,76537	8,01432
3	1,76353	9,00976	-76,88993	0,09345
4	77,92767	34,87079	99,99098	98,98763
5	9,03006	3,33032	4,20009	5,39876
6				

In dieser Tabelle sollen nun folgende (beispielhafte) Berechnungen durchgeführt werden:

- a. In der Tabellenzelle **A6** soll die Summe der Werte in der Spalte **A** berechnet werden.
- b. In der Tabellenzelle **B6** werden die ersten drei Werte aus Spalte **B** addiert und davon die Summe der beiden Werte aus **B4** und **B5** subtrahiert (also $(B1+B2+B3)-(B4+B5)$).
- c. In der Tabellenzelle **C6** soll der Mittelwert aus den Werten der Spalte **C** berechnet werden.
- d. In der Tabellenzelle **D6** soll das Produkt aus dem kleinsten und größten Wert aus dem Zellbereich **A1** bis **D5** (das sind die Tabellenzellen **A1** bis **A5**, **B1** bis **B5**, **C1** bis **C5** und **D1** bis **D5**) berechnet werden.

Für Beispiel **a** gehen Sie wie folgt vor:

1. Setzen Sie die Schreibmarke in die Tabellenzelle **A6**.
2. Wählen Sie im Register **Layout** in der Gruppe **Daten** das Symbol **Formel**. Sie erhalten daraufhin das Dialogfeld **Formel** (siehe Abbildung 1).




Abb. 1: Dialogfeld **Formel**

3. Im Textfeld **Formel** steht (normalerweise) zunächst **=SUM(ABOVE)**. Sie können für dieses Beispiel die vorgegebene Formel belassen, da ja alle Werte „über“ (ABOVE) der aktuellen Tabellenzelle summiert werden sollen. Alternativ können Sie aber auch z.B. **=SUM(A1:A5)** oder auch **=A1+A2+A3+A4+A5** eintragen.
4. Bestätigen Sie das Dialogfeld.

Vom Grundprinzip her können Sie diese vier Schritte nun bei den anderen Beispielen wiederholen. Die nachfolgende Übersicht zeigt Ihnen, wo die Schreibmarke gesetzt werden muss und welche Formel Sie eingeben müssen:

	Schreibmarke	Formel	alternative Formel
b	B6	=(B1+B2+B3)-(B4+B5)	=SUM(B1:B3)-SUM(B4:B5)
c	C6	=AVERAGE(C1:C5)	=SUM(C1:C5)/COUNT(C1:C5)
d	D6	=MIN(A1:D5)*MAX(A1:D5)	=PRODUCT(MIN(A1:D5);MAX(A1:D5))

Zur Kontrolle hier die Ergebnisse der obigen Berechnungen:


A6: 422,87432 **B6: -12,30134** **C6: 7,08317** **D6: -24728,30358**

Wenn nun einige der Ausgangswerte sich zu einem späteren Zeitpunkt ändern, können Sie die neuen, notwendigen Berechnungen schnell und bequem durchführen, ohne die Formeln neu eingeben zu müssen. Markieren Sie einfach die gesamte Tabelle und betätigen die Funktionstaste . Nun werden alle Ergebnisse aktualisiert.

Anmerkung: Falls Sie Angaben in geschweiften Klammern anstelle der erwarteten Ergebnisse sehen (z.B. wird in der Tabellenzelle **A6** angezeigt: `{=SUM(ABOVE)}`), markieren Sie am besten die gesamte Tabelle und verwenden die Tastenkombination  , um die eigentlichen Ergebnisse anzeigen zu lassen. Zeigt die Tastenkombination   keine Wirkung, verwenden Sie die Tastenkombination  .

Zahlenformate festlegen

Neben der Angabe der Formel können Sie im Dialogfeld **Formel** außerdem festlegen, wie viele Stellen das Ergebnis vor bzw. nach dem Dezimaltrennzeichen besitzen soll. Weiterhin können Sie einstellen, dass Punkte als Tausendertrennzeichen oder zusätzliche Textangaben (z.B. **€** oder **EUR**) verwendet werden. Für gängige Zahlenformate gibt es die Liste **Zahlenformat** (siehe Abbildung 2), aus der Sie ein bereits vorhandenes Zahlenformat auswählen können. Sie können aber auch eigene Zahlenformate festlegen. Wird kein Zahlenformat angegeben, wird das Ergebnis mit genau den Nachkommastellen angezeigt, die die Berechnung liefert. Bei den Ziffern vor dem Dezimaltrennzeichen ändert sich nichts. Es wird allerdings auch kein Tausendertrennzeichen angezeigt.



Tragen Sie hier Ihr eigenes Zahlenformat ein ...

... oder wählen ein Zahlenformat in der Liste.

Abb. 2: Zahlenformate im Dialogfeld **Formel**

Die beiden wichtigsten Symbole bei der Angabe eines Zahlenformats sind die Zeichen **0** (Null) und **#**. Beide Zeichen stehen für genau eine Ziffer. Der Unterschied besteht darin, dass bei Verwendung von **#** keine führenden Nullen vor oder nach dem Dezimaltrennzeichen angezeigt werden. Für die

Darstellung von Nachkommastellen wird das Komma (,) als Dezimaltrennzeichen und der Punkt (.) als Tausendertrennzeichen verwendet. Falls bei der Darstellung von positiven und negativen Zahlenergebnissen ein Unterschied gemacht werden soll, können Sie zwei Zahlenformate angeben, getrennt durch ein Semikolon. Das Zahlenformat vor dem Semikolon repräsentiert die positiven, das Zahlenformat hinter dem Semikolon die negativen Zahlen. Falls der ganzzahlige Teil des Ergebnisses einer Berechnung mehr Stellen besitzt, als durch das Zahlenformat angegeben worden ist, werden trotzdem alle Stellen angezeigt. Für die Nachkommastellen gilt: es werden max. so viele Stellen angezeigt, wie im Zahlenformat angegeben wurden. Gegebenenfalls wird auf- bzw. abgerundet. Nachfolgend ein paar Beispiele (das Zeichen _ steht für ein Leerzeichen):

Zahlenformat	Ergebnis der Berechnung	Anzeige im Dokument
###	-4 123 54387 45,7396 0	_ -4 123 54387 _ 46 _ _ _
000	-4 123 54387 45,7396 0	-004 123 54387 046 000
##0	-4 123 54387 45,7396 0	_ -4 123 54387 _ 46 _ _ 0
#.##0	-4 123 54387 45,7396 0	_ _ -4 _ 123 54.387 _ _ 46 _ _ _ 0
#.##0,##	-4 123 54387 45,7396 0	_ _ -4, _ _ _ 123, _ _ 54.387, _ _ _ 45,74 _ _ _ 0, _ _
#.##0,00	-4 123 54387 45,7396 0	_ _ -4,00 _ 123,00 54.387,00 _ 45,74 _ _ _ 0,00
#.##0,00;[#.##0,00]	-4 123 54387 45,7396 0	[_ _ -4,00] _ 123,00 54.387,00 _ 45,74 _ _ _ 0,00



Zahlenformat	Ergebnis der Berechnung	Anzeige im Dokument
#.##0,00 €;[#.##0,00 €]	-4 123 54387 45,7396 0	[_ _ _ 4,00 _ €] _ 123,00 _ € 54.387,00 _ € _ 45,74 _ € _ _ _ 0,00 _ €

Anmerkung: Bei den Berechnungen verwendet Word Feldfunktionen (einschließlich der Angabe des Zahlenformats). Weitere Informationen können Sie im Skript **Word für Microsoft 365 – Feldfunktionen** nachlesen.

Ergebnisse einer Berechnung außerhalb der Tabelle einsetzen

Sie können die Ergebnisse einer Berechnung auch außerhalb einer Tabelle weiterverwenden. Damit auch hier eine automatische Aktualisierung möglich ist, müssen Sie folgende Schritte durchführen:

1. Markieren Sie das Ergebnis in der Tabellenzelle. **Achten Sie darauf, dass Sie nur das Ergebnis markieren, aber nicht die gesamte Tabellenzelle!!!**
2. Wählen Sie im Register **Einfügen** in der Gruppe **Link** das Symbol **Textmarke**. Tragen Sie im Dialogfeld **Textmarke** (siehe Abbildung 3) im Textfeld **Textmarkenname** einen beliebigen Namen ein (z.B. **Rechenergebnis**; der Name darf **keine** Leerzeichen enthalten) und bestätigen das Dialogfeld.



Tragen Sie hier einen benutzerdefinierten Namen für die Textmarke ein. Der Name muss mit einem Buchstaben beginnen, darf Ziffern enthalten, aber keine Leerzeichen (als Trennzeichen ist der Unterstrich _ erlaubt). Länge: max. 40 Zeichen.

Abb. 3: Dialogfeld **Textmarke**

1. Setzen Sie die Schreibmarke an die Textposition außerhalb der Tabelle, wo Sie das Ergebnis der Berechnung einfügen wollen. Wählen Sie im Register **Einfügen** in der Gruppe **Link** das Symbol **Querverweis**. Im Dialogfeld **Querverweis** (siehe Abbildung 4, Seite 8) wählen Sie bei **Verweistyp** den Eintrag **Textmarke**. In der Liste **Für welche Textmarke** wählen Sie den Textmarkennamen, den Sie in Schritt 2 im Dialogfeld **Textmarke** festgelegt haben. In der Liste **Verweisen auf** wählen Sie den Eintrag **Textmarkeninhalte**. Bestätigen Sie das Dialogfeld und an der aktuellen Textposition wird das Ergebnis der Berechnung aus der Tabelle eingefügt.






Abb. 4: Dialogfeld **Querverweis**

Ein großer Vorteil dieser Methode ist der, dass bei einer Änderung des Ergebnisses in der Tabelle (wenn sich die Ausgangswerte für die Berechnung ändern), die Änderung auch auf den Zahlenwert außerhalb der Tabelle überträgt. Dies geschieht übrigens immer automatisch, wenn das Dokument geöffnet oder gedruckt wird.

Logische Funktionen

Neben „einfachen“ Funktionen zur Berechnung von Zahlen, gibt es auch einige logische Funktionen, mit denen Ergebnisse überprüft werden können. Je nachdem, wie das Ergebnis der Überprüfung ausfällt, werden unterschiedliche Informationen angezeigt. Hierzu ein Beispiel:

Nehmen wir an, Sie haben eine Rechnung für einen bestimmten Kunden erstellt. In dieser Rechnung sind die Preise der einzelnen Artikel, die Zwischensumme, der Mehrwertsteuerbetrag und der Gesamtbetrag angegeben. In Tabelle 2a sehen Sie dieses Beispiel.

<i>Anz.</i>	<i>Artikel</i>	<i>Einzelpreis</i>	<i>Gesamtpreis</i>
13	Tintenpatrone für HP DeskJet 540	32,43 €	
4	Toner-Kassette HP-LaserJet 6P	105,00 €	
20	Druckerpapier schmal 2000 Blatt/weiß	27,00 €	
	Zwischensumme:		
	Mehrwertsteuer:	19%	
	Gesamtbetrag:		

Tabelle 2a

Als erstes müssen nun die *Gesamtpreise* für die einzelnen Artikel ermittelt werden. Diese werden dann für die *Zwischensumme* summiert. Aus der *Zwischensumme* müssen Sie den Wert für die *Mehrwertsteuer* errechnen und zum Schluss *Zwischensumme* und *Mehrwertsteuer* zum *Gesamtbetrag* addieren. In Tabelle 2b, Seite 9, sehen Sie dasselbe Beispiel. Diesmal sind jedoch die Formeln in die entsprechenden Tabellenzellen eingetragen. Wie Sie diese Formeln eingeben, haben Sie ja bereits weiter oben kennen gelernt.


Anz.	Artikel	Einzelpreis	Gesamtpreis
13	Tintenpatrone für HP DeskJet 540	32,43 €	=A2*C2
4	Toner-Kassette HP-LaserJet 6P	105,00 €	=A3*C3
20	Druckerpapier schmal 2000 Blatt/weiß	27,00 €	=A4*C4
	Zwischensumme:		=SUM(ABOVE)
	Mehrwertsteuer:	19%	=D6*C7
	Gesamtbetrag:		=D6+D7

Tabelle 2b

Als Zahlenformate sollten Sie für alle Berechnungen **###0,00 €;(###0,00 €)** aus der Liste **Zahlenformat** im Dialogfeld **Formel** wählen.

Nun zur eigentlichen Aufgabe: in Abhängigkeit von der Größe des Gesamtbetrags legen Sie die Höhe des Skontos fest, dass Sie bei Einzahlung in einem gewissen Zeitraum gewähren wollen. Als Beispiel gewähren Sie bei einem Gesamtbetrag unter 1.000,00 € ein Skonto von 2%, ab dem Wert von 1.000,00 € ein Skonto von 3%. Diese Angabe werden Sie i. Allg. in einem Absatz machen der unterhalb, und damit außerhalb der Tabelle steht. Hierfür sind folgende Schritte notwendig:

1. Markieren Sie den Gesamtbetrag in der Tabelle (das ist die Tabellenzelle mit der Formel **=D6+D7**), aber **nicht** die gesamte Tabellenzelle.
2. Wählen Sie im Register **Einfügen** in der Gruppe **Link** das Symbol **Textmarke**. Tragen Sie im Dialogfeld **Textmarke** (siehe Abbildung 3, Seite 7) im Textfeld **Textmarkenname** einen beliebigen Namen ein (z.B. **Gesamtergebnis**) und bestätigen das Dialogfeld.
3. Setzen Sie die Schreibmarke an die Textposition außerhalb der Tabelle, wo Sie den Betrag **2%** bzw. **3%** eintragen wollen.
4. Drücken Sie die Tastenkombination **Strg F9**. Es erscheint ein paar geschweiften Klammern **{ }** mit zwei Leerzeichen (hier dargestellt mit dem Zeichen) zwischen den Klammern. Die Schreibmarke befindet sich genau zwischen den beiden Leerzeichen.
5. Geben Sie folgende Anweisung ein (es wird an dieser Stelle davon ausgegangen, dass die Textmarke *Gesamtergebnis* lautet (siehe Schritt 2); die Groß-/Kleinschreibweise ist ohne Bedeutung):



IF_Gesamtergebnis_<_1000_ "2%" "3%"

Achten Sie dabei genau auf die Leerzeichen (hier dargestellt mit dem Zeichen), die nicht fehlen dürfen.

6. Drücken Sie die Funktionstaste **F9** (achten Sie darauf, dass sich die Schreibmarke immer noch zwischen den beiden geschweiften Klammern befindet). Die Darstellung mit den geschweiften Klammern und der eingegebenen Anweisung verschwindet und stattdessen erscheint das Ergebnis der Anweisung (entweder **2%** oder **3%**).

Alle Funktionen im Überblick

Hier nun ein Überblick über alle Funktionen, die Sie im Dialogfeld **Formel** einsetzen können. Für die Beispiele in der dritten Spalte der Auflistung, hier zunächst eine Beispieltabelle:

	A	B	C	D
1	12,6543	-4.567,121	-324,85798	231,754
2	-32,6574	-67,632	87,34533	3.234,997
3	123,9685	90,009	-7.764,87642	501,663
4	4.886,8574	8.764,935	0,00645	-0,092

Funktion	Beschreibung	Beispiel
ABS (mathematisch)	Berechnet den Absolutbetrag einer Zahl, d.h. eine negative Zahl wird in eine positive Zahl umgewandelt.	=ABS(-5) ⇒ 5 =ABS(C1) ⇒ 324,85798
AND (logisch)	Liefert den Wert 1 (wahr), wenn zwei logische Berechnungen jeweils den Wert 1 liefern, ansonsten 0 (falsch).	=AND(3<2;4>3) ⇒ 0 =AND(A1>A2;C1<D3) ⇒ 1
AVERAGE (mathematisch)	Liefert den Mittelwert einer Zahlenreihe.	=AVERAGE(3;2;7;5) ⇒ 4,25 =AVERAGE(A4;B1;C2) ⇒ 135,69391
COUNT (mathematisch)	Liefert die Anzahl der angegebenen Zahlenwerte bzw. des angegebenen Zellbereichs.	=COUNT(1;4;6;3;7) ⇒ 5 =COUNT(A1:C3) ⇒ 9
DEFINED (logisch)	Liefert den Wert 1 (wahr), wenn die angegebene Zelle existiert, sonst 0 (falsch).	=DEFINED(A1) ⇒ 1 =DEFINED(G3) ⇒ 0
FALSE (logisch)	Liefert den Wert 0 und wird mit anderen logischen Funktionen verwendet.	=FALSE ⇒ 0
IF (logisch)	Überprüft eine logische Bedingung und liefert ein Ergebnis, wenn die Bedingung wahr ist (obligatorisch) und ein anderes Ergebnis, wenn die Bedingung falsch ist (optional).	=IF(3<7;A1+10;A1-10) ⇒ 22,6543 =IF(3>7;A1+10;A1-10) ⇒ 2,6543
INT (mathematisch)	Schneidet sämtliche Nachkommastellen ab, dabei wird weder auf- noch abgerundet.	=INT(23,65*9,8756) ⇒ 233 =INT(B4) ⇒ 8764
MAX (mathematisch)	Liefert den größten Wert einer Zahlenreihe.	=MAX(34;65;2;4,5;132) ⇒ 132 =MAX(A1:D2) ⇒ 3234,997
MIN (mathematisch)	Liefert den kleinsten Wert einer Zahlenreihe.	=MIN(34;65;2;4,5;132) ⇒ 2 =MIN(A1:D2) ⇒ -4567,121
MOD (mathematisch)	Liefert den ganzzahligen Rest, der bei einer Division übrigbleibt.	=MOD(45;17) ⇒ 11 =MOD(INT(A3);INT(A1)) ⇒ 3

Funktion	Beschreibung	Beispiel
NOT (logisch)	Negiert das Ergebnis einer logischen Operation.	=NOT(2<3) ⇒ 0 =NOT(C2<C3) ⇒ 1
OR (logisch)	Liefert den Wert 1 (wahr), wenn bei 2 logischen Berechnungen min. eine Berechnung den Wert 1 liefert, sonst 0 (falsch).	=OR(2<3;3>4) ⇒ 1 =OR(C1>C3;D1<B2) ⇒ 1
PRODUCT (mathematisch)	Berechnet das Produkt einer Zahlenreihe.	=PRODUCT(3;4;8;12) ⇒ 1152 =PRODUCT(C3:D4) ⇒ 2311,5014
ROUND (mathematisch)	Rundet das Ergebnis auf eine anzugebende Zahl von Nachkommastellen auf bzw. ab.	=ROUND(23,6387;3) ⇒ 23,639 =ROUND(C2;2) ⇒ 87,35
SIGN (logisch)	Liefert den Wert -1 , wenn das Ergebnis einer Berechnung negativ ist, sonst 1 .	=SIGN(3*-4) ⇒ -1 =SIGN(D1) ⇒ 1
SUM (mathematisch)	Berechnet die Summe einer Zahlenreihe.	=SUM(3;6;12;76;89) ⇒ 186 =ROUND(SUM(C1:D3);3) ⇒ -4033,975
TRUE (logisch)	Liefert immer den Wert 1 und wird mit anderen logischen Funktionen verwendet.	=TRUE ⇒ 1

Anmerkung: Für einige Funktionen (u.a. **SUM**, **PRODUCT**) können Sie als Funktionsargument **ABOVE** angeben (siehe auch Abbildung 1, Seite 4), wenn alle Werte in der Spalte oberhalb der Tabellenzelle mit der Formel für die Berechnung verwendet werden sollen. Neben **ABOVE** können Sie aber auch **RIGHT**, **LEFT** oder **BELOW** angeben, wenn die Werte rechts, links oder unterhalb der Tabellenzelle mit der Formel für die Berechnung verwendet werden sollen.

Komplexe Funktionen in Tabellen verwenden

Wenn Sie Funktionen für komplexere Berechnungen in einer Tabelle verwenden möchten, gibt es die Möglichkeit, die Funktionen des Tabellenkalkulationsprogramms Excel einzusetzen. Voraussetzung ist allerdings, dass Sie Excel auf Ihrem PC installiert haben. Um eine Excel-Tabelle direkt in Word erstellen zu können, gehen Sie wie folgt vor:

1. Setzen Sie die Schreibmarke an die Textstelle (am besten eignet sich ein leerer Absatz), wo die Excel-Tabelle eingefügt werden soll.
2. Wählen Sie im Register **Einfügen** in der Gruppe **Tabelle** das Symbol **Tabelle** und den Befehl **Excel-Kalkulationstabelle**. Es wird eine Excel-Tabelle mit einer zunächst festen Anzahl von Spalten und Zeilen eingefügt.
3. Um das Excel-Tabellenfenster herum befindet sich ein Rahmen mit acht kleinen, schwarzen Quadraten (vier an den Mittelpunkten der Seiten und vier in den Ecken). Über die Quadrate können Sie mit der Maus die genaue Anzahl an Spalten und Zeilen festlegen. In Word werden nachher genau die Tabellenzellen angezeigt, die auch jetzt im Excel-Tabellenfenster zu sehen sind. Das bedeutet, Sie müssen evtl. das Excel-Tabellenfenster verkleinern, wenn Sie nicht alle Tabellenzellen mit Daten oder



Funktionen ausfüllen, oder vergrößern, falls die angezeigte Anzahl an Spalten bzw. Zeilen nicht ausreichen sollte.

4. Tragen Sie in die Tabellenzellen die Tabelle ein. Für Berechnungen stehen Ihnen alle Excel-Funktionen zur Verfügung (das Menüband ist das von Excel).
5. **Optional**: Formatieren Sie die Tabelle mit den Formatierungsmöglichkeiten von Excel. Die Excel-Formatierungen werden in Word übernommen.
6. Wenn die Excel-Tabelle fertig ist, klicken Sie mit der Maus auf eine beliebige Stelle im Word-Dokument (außerhalb des Excel-Tabellenfensters) und Sie haben die fertige Excel-Tabelle in Ihrem Word-Dokument.

Im Word-Dokument können Sie nun nicht wirklich viel mit dieser Excel-Tabelle anfangen (u. a. können Sie die Tabelle von der Breite oder Höhe her anpassen). Wollen Sie in der Tabelle selbst Änderungen vornehmen (z.B. sollen Daten oder die Formatierung geändert werden), bewegen Sie das Mausymbol auf die Excel-Tabelle und machen einen Doppelklick. Sie befinden sich nun wieder im Excel-Tabellenfenster und können erneut die Möglichkeiten von Excel nutzen. Zur Rückkehr nach Word wieder einfach eine beliebige Textstelle im Word-Dokument anklicken. Abbildung 5 zeigt ein Beispiel für eine solche Excel-Tabelle im Word-Dokument.




Abb. 5: Excel-Tabelle im Word-Dokument, in der Bearbeitung und nach der Bearbeitung